
b What are Invariants? c

• A Term is either a variable symbol, a quoted
constant, or a function application

− Example:
(cons (binary-+ x (quote 1)) ’(t . nil))

◦ Every function is either a function symbol or a
lambda expression

• A Predicate is a term with a single variable
symbol n and is interpreted in an iff context

− This is our non-standard definition of Predicate

• An Invariant is a predicate which we wish to
prove is non-nil for all values of n.

− The variable n is intended to range over all values of
natural-valued “time”

b Importance of Proving Invariants c

• Most properties of interest about concurrent,
reactive systems can be effectively reduced to
the proof of a sufficient invariant

• Invariants can be very difficult and tedious to
prove for larger systems.

− Many examples of this phenomenon from the ACL2
community and other formal methods communities

b Example Invariant: Mutual Exclusion c

(encapsulate (((i *) => *))

(local (defun i (n) n)))

(define-system mutual-exclusion

(in-critical (n) nil

(if (in-critical n-)

(/= (i n) (critical-id n-))

(= (status (i n) n-) :try)))

(critical-id (n) nil

(if (and (not (in-critical n-))

(= (status (i n) n-) :try))

(i n)

(critical-id n-)))

(status (p n) :idle

(if (/= (i n) p) (status p n-)

(case (status p n-)

(:try (if (in-critical n-)

:try

:critical))

(:critical :idle)

(t :try)))))

b Specifying Mutual Exclusion c

• Property: No two distinct processes a and b
can be in the :critical state at the same time

• Codified as the invariant (ok n):

(encapsulate (((a) => *) ((b) => *))

(local (defun a () 1))

(local (defun b () 2))

(defthm a-/=-b (not (equal (a) (b)))))

(defun ok (n)

(not (and (= (status (a) n) :critical)

(= (status (b) n) :critical))))

b Approaches - Theorem Proving c

•Define and prove an inductive invariant which
implies the target invariant.

− For complex systems, the definition and/or proof of
an inductive invariant is a non-trivial exercise

• For our mutual exclusion example:

(defun ii-ok-for1 (n i)

(iff (= (status i n) :critical)

(and (in-critical n)

(= (critical-id n) i))))

(defun ii-ok (n)

(and (ii-ok-for1 n (a)) (ii-ok-for1 n (b))))

(defthm ii-ok-is-inductive-invariant

(and (ii-ok (t0))

(implies (ii-ok n)

(and (ok n) (ii-ok (t+ n))))))

(defthm ok-is-invariant (ok n))

b Approaches - Model Checking c

• Explore an “effective” finite state graph of a
system searching for failures

− Specification is usually provided by a temporal logic
formula: e.g. an invariant in CTL would be AG(ok)

− System definition languages: Verilog HDL, VHDL,
SMV, Murφ, SPIN, Limited variants of C/C++, etc.

− Model checkers are generally classified into explicit-
state and implicit-state

− Several algorithms exist to reduce large-state systems
to effectively finite abstract state systems: symmetry
reductions, partial order reductions, etc.

• Hybrid approaches: too many to enumerate,
but most involve some form of abstraction.

b Our Approach - Phase 1 c

• Assume the definition of a term rewrite func-
tion rewrt which takes a term as an input and
produces the rewritten term

• For a predicate φ, denote φ′ as the term:
(rewrt ‘((lambda (n) ,φ) (t+ n)))

• Assume the following function definition:

(defun state-ps (trm)

(cond ((or (atom trm) (quotep trm)) ())

((eq (first trm) ’if)

(union-equal (state-ps (second trm))

(union-equal (state-ps (third trm))

(state-ps (fourth trm)))))

(t (and (state-predp trm) (list trm)))))

• Compute the least set of predicates Ψ s. t. :
(a) the target invariant predicate τ ∈ Ψ, and
(b) for every φ ∈ Ψ, (state-ps φ′) ⊆ Ψ

b Our Approach - Phase 2 c

• From the φ′, we compute a finite set of input
(non-state) predicates Γ

− For each predicate α in Ψ∪Γ, define a boolean variable
bv(α)

• For each φ in Ψ, we replace the predicate sub-
terms α in φ′ with bv(α)

− This gives us a next-value function for computing
the next value of bv(φ) in terms of the current values of
the boolean variables

• Explore the abstract graph defined by the
next-value functions for bv(Ψ)

− nodes in the graph are valuations of the variables
bv(Ψ) and an edge exists from one node to the next if a
valuation of bv(Γ) exists

− If no path is found to a node where bv(τ ) is nil, then
return Q.E.D.

− Otherwise, return a pruned version of the failing path
to the user for further analysis

b Our Approach - Elaborations c

• The function (state-predp trm) is essen-
tially defined as:

(defun state-predp (trm)

(and (not (intersectp-eq (all-fnnames trm) ’(t+ hide)))

(equal (all-vars trm) ’(n))))

− Thus, the user can introduce an input predicate by
introducing a hide

• We chose to define our own term rewriter for
numerous reasons

− The rewriter does extract rewrite rules from the cur-
rent ACL2 world

• Our “model checker” is a compiled, optimized
(to an extent), explicit-state, breadth-first search
through the abstract graph

• The prover also supports assume-guarantee
reasoning through the use of forced hypothesis

b Mutual Exclusion Continued c

• Beginning with τ = (ok n), the prover gener-
ates the following set of predicates Ψ:

(ok n)

(equal (status (a) n) ’:critical)

(equal (status (b) n) ’:critical)

(equal (status (a) n) ’:try)

(equal (status (b) n) ’:try)

(in-critical n)

(equal (critical-id n) (a))

(equal (critical-id n) (b))

• The resulting abstract graph has 20 nodes and
verifies that (ok n) is never nil

• We can further reduce the graph to 6 nodes
by hiding :try terms:

(defthm coerce-try-status-to-input

(equal (equal (status p n) ’:try)

(hide (equal (status p n) ’:try))))

b ESI cache example-1 c

• Another example: a high-level definition of
the ESI cache coherence protocol

• System defined by following state variables:

− (mem c n) – shared memory data for cache-line c

− (cache p c n) – data for cache-line c at proc. p

− (valid c n) and (excl c n) – sets of processor
id.s which define the ESI cache states

• We will need a few constrained functions:

(encapsulate (((proc *) => *) ((op *) => *)

((addr *) => *) ((data *) => *))

(local (defun proc (n) n)) (local (defun op (n) n))

(local (defun addr (n) n)) (local (defun data (n) n)))

(encapsulate (((c-l *) => *)) (local (defun c-l (a) a)))

(define-system mesi-cache

(mem (c n) nil

(cond ((/= (c-l (addr n)) c) (mem c n-))

((and (= (op n) :flush)

(in1 (proc n) (excl c n-)))

(cache (proc n) c n-))

(t (mem c n-))))

(cache (p c n) nil

(cond ((/= (c-l (addr n)) c) (cache p c n-))

((/= (proc n) p) (cache p c n-))

((or (and (= (op n) :fill) (not (excl c n-)))

(and (= (op n) :fille) (not (valid c n-))))

(mem c n-))

((and (= (op n) :store) (in1 p (excl c n-)))

(s (addr n) (data n) (cache p c n-)))

(t (cache p c n-))))

(excl (c n) nil

(cond ((/= (c-l (addr n)) c) (excl c n-))

((and (= (op n) :flush)

(implies (excl c n-)

(in1 (proc n) (excl c n-))))

(sdrop (proc n) (excl c n-)))

((and (= (op n) :fille) (not (valid c n-)))

(sadd (proc n) (excl c n-)))

(t (excl c n-))))

(valid (c n) nil

(cond ((/= (c-l (addr n)) c) (valid c n-))

((and (= (op n) :flush)

(implies (excl c n-)

(in1 (proc n) (excl c n-))))

(sdrop (proc n) (valid c n-)))

((or (and (= (op n) :fill) (not (excl c n-)))

(and (= (op n) :fille) (not (valid c n-))))

(sadd (proc n) (valid c n-)))

(t (valid c n-)))))

b ESI cache example-3 c

• Property: the value read by a processor is the
last value stored.

• A codification in ACL2 of this property as the
target invariant (ok n):

(encapsulate (((p) => *) ((a) => *))

(local (defun p () t)) (local (defun a () t)))

(define-system mesi-specification

(a-dat (n) nil

(if (and (= (addr n) (a))

(= (op n) :store)

(in1 (proc n) (excl (c-l (a)) n-)))

(data n)

(a-dat n-)))

(ok (n) t

(if (and (= (proc n) (p))

(= (addr n) (a))

(= (op n) :load)

(in (p) (valid (c-l (a)) n-)))

(= (g (a) (cache (p) (c-l (a)) n-)) (a-dat n-))

(ok n-))))

b ESI cache example-4 c

• Key rewrite rule to introduce case splits on
the exclusive set (excl c n):

(defthm in1-case-split

(equal (in1 e s)

(cond ((not s) nil)

((c1 s) (equal e (scar s)))

(t (hide (in1 e s))))))

• Prover generates following predicate set and
explores resulting graph (11 nodes):

(ok n)

(valid (c-l (a)) n)

(in (p) (valid (c-l (a)) n))

(excl (c-l (a)) n)

(c1 (excl (c-l (a)) n))

(equal (scar (excl (c-l (a)) n)) (p))

(equal (a-dat n) (g (a) (mem (c-l (a)) n)))

(equal (a-dat n) (g (a) (cache (p) (c-l (a)) n)))

(equal (a-dat n) (g (a) (cache (scar (excl (c-l (a)) n))

(c-l (a)) n)))

b Conclusions and Future Work c

• Prover can be effective but requires thought:

− Careful consideration of system definition and speci-
fication relative to existing operators and rewrite rules

− Determination of which terms should be hidden and
the possible addition of auxiliary variables

• Improvements to the Prover:

− Interfaces to external model checkers for Phase 2

− Better methodology for prover use and user feedback

• Many more example systems and effort to
integrate with RTL definitions and existing li-
brary

• Need to develop more comprehensive compo-
sitional methodology

